
Multiple CPUs, matrix computations
Parallel loops

Syntax frames

Below a series of slides on syntax in Ox.
Read them through, try out in small programs if you understand
the meaning.

Getting started

Chapter 1: Getting started
Exercise:

1. Copy the file <ox-home>/samples/myfirst.ox to your
personal directory.

2. Open the file in OxEdit (e.g. Windows Explorer, walk there,
right mouse button, Send To - OxEdit)

3. Run the program (through Modules - Run - Ox)
(If there is no Ox option under the Run menu, load the OxPro.tool file from the lists/tool directory, using Tools -

Add/remove modules - Load from)

Output

Ox Professional version 7.00 (Linux_64/MT) (C) J.A. Doornik , 1994 -2012

two matrices

2.0000 0.0000 0.0000

0.0000 1.0000 0.0000

0.0000 0.0000 1.0000

0.0000 0.0000 0.0000

1.0000 1.0000 1.0000

Getting started

Using OxEdit
One tab has program
Running the program puts output in separate file/sheet
Errors in code can appear in output file
Workspace indicates opened files

Type of errors
1. Compilation errors: Like the above, error in the syntax of Ox

Listing 1: myfirst err.ox
print "two matrices", m1, m2);

// gives compile -time error

----------------

Ox Professional version 7.00 (Linux_64/MT) (C) J.A. Doornik , 1994 -2012

myfirst_err.ox (12): ’;’ expected but found ’<string >’

myfirst_err.ox (12): ’;’ expected but found ’)’

myfirst_err.ox (12): ’)’ out of place

2. Runtime errors: Impossible computations or commands

Listing 2: myfirst err.ox
print ("product of two matrices", m1 * m2);

// gives run -time error

----------------

Ox Professional version 7.00 (Linux_64/MT) (C) J.A. Doornik , 1994 -2012

...

Runtime error: ’matrix [3][3] * matrix [2][3] ’ bad operand

Runtime error occurred in main (14), call trace:

myfirst_err.ox (14): main

One error can lead to multiple messages: Start solving first in list.

Chapter 2: Syntax - Comments

/* This is standard comment ,

which /* may be nested */.

*/

decl x; // declare the variable x

Use them well, use them extensively, use them consistently

/*

** olsc(const mY , const mX , const amB)

**

** Purpose:

** Performs OLS , expecting the data in columns.

**

** Inputs:

** mY iT x iN matrix of regressors Y

** mX iT x iK matrix of explanatory variables X

**

** Outputs:

** amB address of iK x iN matrix with iN sets of OLS coefficients

**

** Return value:

** integer , 1: success , 2: rescaling advised ,

** -1: X’X is singular , -2: combines 2 and -1.

**

** Example:

** ir = olsc(mY , mX , &mB);

**

** Last changed

** 21 -04 -96 (Marius Ooms ): made documentation

** 06 -08 -09 (Charles Bos ): adapted documentation

*/

Use explanation, consistently, before every function, detailing
name, purpose, inputs, outputs, return value (and possibly date,
author, once per file)

Program layout
A minimal complete program is:

Listing 3: oxtut2b.ox
#include <oxstd.h>

main()

{

println("Hello world");

}

Contains:

1. Include statement, to define all standard functions in Ox;
between < and > to indicate oxstd.h is an intrinsic part of
Ox

2. One function header, called main, taking no arguments ()

3. Function body for main(), enclosed in {}, with a println

statement

Note: Syntax terribly similar to C or Java.

Statements
Listing 4: oxtut2c-hun.ox

#include <oxstd.h>

main()

{

decl iN, dSigma , mX, vBeta , vEps;

iN = 4;

dSigma = 0.25;

mX = 1 ∼ ranu(iN , 2);

vBeta = <1; 2; 3>;

vEps = dSigma * rann(iN, 1);

print("x", mX, "beta", vBeta , "epsilon", vEps);

}

(note: Stick to Hungarian, don’t follow the Introduction to Ox
literally here)

I Declaration: Automatic typing
I Assignment: Integer, double, matrix-function,

matrix-constant, function result.
I Print statement

Syntax

Identifiers, variables, matrices

Identifiers

Identifiers: All names of variables, constants and functions

1. Case sensitive

2. Distinct between blocks of the program; local declaration can
overrule global declaration

3. Contain [A-Z], [a-z], [0-9], [ ], and start with a letter.

4. Do use sensible names; use Hungarian notation for your own
sake

Syntax

Identifiers, variables, matrices

Matrix creation I

I <1, 2, 3> creates a row vector

I <1.1; 2.2; 3.3> creates a column vector

I <0, 1, 2; 3, 4, 5> creates a 2× 3 matrix

I <1:4> is the same matrix as <1, 2, 3, 4>

I You cannot combine a matrix constant with a variable:
<1, 2, dSigma> leads to a compilation error

Identifiers, variables, matrices

Matrix creation II
I Assign a matrix constant mX= <1, 2>;
I Assign another matrix or function of matrices mX= mY + mZ;
I Assign the result of a standard function,

mX= unit(2); mY= zeros(2, 6); mZ= range(0, 1, .05);
I Concatenate other elements

mX= 1~mY; mZ= mX|mY, mY= (0~1)|(2~3);

Check that the matrices ‘fit’ when you concatenate or sum.
Scalars fit everywhere.

Warning: Concatenating matrices is (relatively) slow, don’t do it
within a loop. Compare:

Listing 5: inefficient
mX= <>;

for (i= 0; i < 1000; ++i)

// Concatenate random numbers

mX|= rann(1, 5);

Listing 6: efficient
mX= zeros (1000, 5);

for (i= 0; i < 1000; ++i)

// Place random numbers

mX[i][]= rann(1, 5);

Functions

Simple functions
The most simple Ox function has no arguments, and returns no
value. The syntax is:
function name ()
{

statements
}

For example:

Listing 7: func-sometext.ox
#include <oxstd.h>

sometext ()

{

print("Some text\n");

}

main()

{

sometext ();

}

Syntax

Functions

Function arguments

I Each function can take one or more arguments.

I [Each argument can be declared const, or non-constant. For non-constant arguments, Ox copies the value

of the argument internally, and hence it is slower than using const arguments.]

I [Ox7 is smarter: Forgetting about const is no longer slower. But less robust, as you might accidentally

change a local variable without lasting effect.]

I ⇒ Always declare your arguments to be const.

dimensions(const mX)

{

println("The argument has ",

rows(mX), " rows and ", columns(mX), " columns.");

}

main()

{

dimensions(zeros (40, 5));

}

Syntax

Functions

Function arguments: Default/variable

I (The lasts arguments may have a default declaration, where
default declaration is filled in where needed. Advanced, but
useful)

I (The last argument may be a set of three dots, ..., indicate a
variable number of arguments. Advanced, sometimes useful,
see ox_vararg.ox)

sometext(const sText="Hello world")

{

println("The argument sText= ", sText);

}

main()

{

sometext ();

sometext("... again");

}

Functions

Forward function declarations (ugly...)

Ox can use a function only when it is known, or at least when the
calling sequence is known. Hence either

1. Put the functions before the main() routine

2. Put the function after the main() routine, and use a forward
declaration, putting the function heading with a semicolon up
front.

MyOls(const mY , const mX); // forward declaration

main()

{

// Now MyOls may be used here

}

MyOls(const mY , const mX)

{

// Specification of MyOls

}

The header files (e.g. oxstd.h) mainly list all the function
declarations together, whereas the source code resides elsewhere.

Syntax

Output from functions

Returning a value

The syntax of the return statement is:
return return value ;

Or, to exit from a function without a return value:
return;

You may exit from a function at the end, or also at an earlier
stage; remaining commands are not executed.
If you exit at the end, and do not want to return anything, return
statement is not needed.

Syntax

Output from functions

Multiple returns

Multiple values can be returned as an array:

func()

{

return { mA, sB, vC };

}

which can then be assigned as follows:

[mX , sY , vZ] = func ();

Note how the names within the routine should match, and the
names outside the routine (e.g. in the main() routine) should
match; what is called mX in main() can be called mA in func.

20/66



PPEctr-Syntax

Syntax

Output from functions

Returning values through arguments
Quite often more convenient to call a routine such that an
argument can get changed, e.g.

ir= MyOlsc(vY, mX, &vBeta);

I This call passes an address of vBeta to MyOlsc

I The address itself is not changed in MyOlsc

I Only what is at the address [color of building], is changed

Listing 8: myolsc.ox
MyOlsc(const vY , const mX, const avBeta)

{

// Adapt the value at the address avBeta , its first array value

avBeta [0]= invertsym(mX’mX)*mX’vY;

return 1;

}

Syntax

Output from functions

Checking arguments

Listing 9: oxtut2g hun.ox
#include <oxstd.h>

test1(iX) // no const , because x will be changed

{

iX = 1;

println("in test1: x=", iX);

}

test2(const aiX)

{

// Change value AT address , not the address itself

aiX [0] = 2;

println("in test2: x=", aiX [0]);

}

main()

{

decl iX;

iX= 10;

println("x = ", iX);

test1(iX); // pass x

println("x = ", iX);

test2(&iX); // pass address of x

println("x = ", iX);

}

Syntax

Indexing

Indexing

All items with multiple components can be indexed.

Note that indexing starts at 0, as in C/C++

I mX[0][1]: Element in the first row, second column of matrix
mX

I mX[][i]: All elements of column i + 1

I mX[3:4][i:j]: The submatrix from rows 4-5 and columns
i + 1 to j + 1.

I mX[:2][]: The first three rows of the matrix

I mX[miI][miJ]: Advanced: The cross-section of rows with
indices in miI and columns with indices in miJ are given.

Syntax

Indexing

Indexing II

Other indexing

I sName[3:6]: Letters 4-7 of a string

I sName[3]: The (integer) ASCII value of letter 4 of a string!

I avX[2]: Element 3 of an array; according to the Hungarian
notation of the name, this seems to be a vector.

I amX[2][0][1]: Element in the first row, second column, of
the matrix at element 3 of the array. Matrices are
2-dimensional, further dimensions implemented as arrays.

24/66



PPEctr-Syntax

Syntax

Indexing

Indexing II

Other indexing

I sName[3:6]: Letters 4-7 of a string

I sName[3]: The (integer) ASCII value of letter 4 of a string!

I avX[2]: Element 3 of an array; according to the Hungarian
notation of the name, this seems to be a vector.

I amX[2][0][1]: Element in the first row, second column, of
the matrix at element 3 of the array. Matrices are
2-dimensional, further dimensions implemented as arrays.

24/66



PPEctr-Syntax

Syntax

Indexing

Indexing II

Other indexing

I sName[3:6]: Letters 4-7 of a string

I sName[3]: The (integer) ASCII value of letter 4 of a string!

I avX[2]: Element 3 of an array; according to the Hungarian
notation of the name, this seems to be a vector.

I amX[2][0][1]: Element in the first row, second column, of
the matrix at element 3 of the array. Matrices are
2-dimensional, further dimensions implemented as arrays.

24/66



PPEctr-Syntax

Syntax

Indexing

Indexing II

Other indexing

I sName[3:6]: Letters 4-7 of a string

I sName[3]: The (integer) ASCII value of letter 4 of a string!

I avX[2]: Element 3 of an array; according to the Hungarian
notation of the name, this seems to be a vector.

I amX[2][0][1]: Element in the first row, second column, of
the matrix at element 3 of the array. Matrices are
2-dimensional, further dimensions implemented as arrays.

24/66



PPEctr-Syntax

Syntax

Operators

Operators

operator operation

’ transpose, m × n m × k n × k A′b
^ (matrix) power m ×m 1× 1 m ×m Ab

* (matrix) multiplication n × k k ×m n × k AB
/ (matrix) division m × n p × n p ×m AB−1

** (matrix) Kronecker product m × n p × q mp × nq aijB
+ addition m × n m × n m × n A+ B
- subtraction m × n m × n m × n A− B
~ horizontal concatenation m × n m × k m × n + k [A B]
| vertical concatenation m × n k × n m + k × n [A;B]

.^ element-by-element power m × n 1× 1 m × n abij

.* element-by-element multiplication m × n m × n m × n aijbij

./ element-by-element division m × n m × n m × n aij/bij

Operators

Operators: Special cases

I A scalar combines with everything. Correct is 1~mX

(concatenate a vector of ones with mX), incorrect is <1>~mX

(unless mX has one row; it results in a warning that the
matrix is padded with zeros to make things fit).

I Adding (or subtracting) a row and column vector is correct:

(
x0 x1

)
+

 y0

y1

y2

 =

 x0 + y0 x1 + y0

x0 + y1 x1 + y1

x0 + y2 x1 + y2

 .

Operators

Relational and logical operators

Comparison can be done in two ways

I Using the standard operators: Results in one, scalar, outcome,
either TRUE≡ 1 or FALSE≡ 0. Note that e.g. mX > mY is true
only when all elements of mX are larger than the corresponding
elements of mY

I Dot-version: Using element-by-element operators results in a
matrix filled with 0’s and 1’s.

Relational operators
operator dot-version operation

< .< less than
> .> greater than
<= .<= less than or equal to
=> .=> equal or greater than
== .== is equal
!= .!= is not equal

Logical operators
operator dot-version operation

&& .&& logical-and
|| .|| logical-or

Operators

Comments on operators

Question

If (mX < mY) = FALSE, then what is the outcome of the
comparison (mX >= mY)?

Boolean shortcut
If an expression involves several logical operators after each other,
evaluation will stop as soon as the final result is known. For
example in (1 || checkval(mX)) the function checkval is
never called, because the result will be true regardless of its
outcome. This is called a boolean shortcut.

Assignments

Assignments and combinations

Assignment is also an operator, i.e., an assignment ‘leaves a value’
which can be used in further assignments:

decl x1, x2, x3, x4;

x1= 0; x2= 0; x3= 0; x4= 0;

// or more concisely

x1= x2= x3= x4= 0;

Some others:
x1+= 2; x4/= (x1+x2);

x2 -= x1; x1∼= x2;

x3*= 5; x4|= x3;

++x1; --x2;

x1++; x2++;

Assignments

Quiz-question: 5-minute exercise

Check in a small program the difference between ++x1 and x1++.

Who is the first to find it?

Assignments

Conditional assignment

Advanced, but useful shortcut

Listing 10: oxcond.ox
if (dX > 0)

dY = 1;

else

dY = -1;

// is equivalent to

dY= (dX > 0) ? 1 : -1;

Can also be done element-by-element, i.e.

mY= (mX .> 0) .? 1 .: -1;

would create a matrix mY of the same size of mX, containing 1, -1
according to the sign of mX.
Very useful in creating dummies, think of probit models.

Combining assignments: Comma operator (ugly...)

One statement runs from a ; to the next ;.
One statement may contain multiple assignments, split by the
comma operator:

i= 1, k=2;

You might just as well put

i= 1; k=2;

in most situations; using the comma operator is ugly in most
situations. A possible exception is in the initialisation of a for-loop:

decl i, k;

for (i= 0, k= 1; i < 5; i += 2)

print ("i= ", i, " k= ", k);

// Easier to read is the following

k= 1;

for (i= 0; i < 5; i += 2)

print ("i= ", i, " k= ", k);

Operator precedence

See table 3.1 in the introduction, of the web-page on your
computer.
Be careful at first, use parentheses to make sure.

Loops

For-loops
At a later stage, we discuss looping constructs in more detail. For
the exercise, you need the for-loop.
Syntax:
for (init_expr; test_expr ; increment_expr)

statement

Steps in the for-loop are

1. Initialise, executing the init_expr
2. If the test_expr is true

2.1 execute the statement,
2.2 execute the increment_expr, and go to 2.

3. Continue with first statement after the loop.

The statement can either be a singular statement, e.g.

dX= rann(1, 5);

or a compound statement, blocking together a group of statements
within curly parentheses { }.

Loops

Example for-loop

Listing 11: oxforloop.ox
k= 1;

for (i= 0; i < 5; ++i)

{

k*= 2;

println ("i= ", i, " k= ", k);

}

What would be the output?

Loops

Loop: For

See earlier frames. More extensive example

Listing 12: oxforloop ext.ox
#include <oxstd.h>

main()

{

decl i, k;

for (i= 0, k= 1; (i < 5) && (k < 7); ++i, k*= 2)

println ("i= ", i, " k= ", k);

...

}

The initialisation and increment statements can be split into many
segments separated by comma’s; the test statement can be a
compounded test.
For your own sake: Don’t follow the example, keep the loop
simple, e.g. use a while-loop instead.

Loops

Loop: While

Listing 13: oxforloop ext.ox
println ("With a while -loop");

i= 0; k= 1;

while ((i < 5) && (k < 7))

{

println ("i= ", i, " k= ", k);

++i;

k*= 2;

}

or, to run the loop at least once:

i= 0; k= 1;

do

{

println ("i= ", i, " k= ", k);

++i;

k*= 2;

}

while ((i < 5) && (k < 7));

Loop: foreach

Alternatively, loop over the elements of a matrix, array, or string.
One can get access to the index.

Listing 14: oxforeachloop.ox
decl i, vX, dX;

vX= <5, 2, 8>;

foreach (dX in vX[i])

{

println ("i= ", i, ", x= ", dX);

}

Note how the index get’s its value ‘magically’, without direct
assignment.
This also works for looping over the elements of e.g. mX[i][j], or
an array asX[i]. There is no need to use the index.

Conditional statements

Conditional statements: If

if ( condition )

statement

else if ( condition )

statement

else

statement

A condition evaluating to a non-zero value is considered true. For
a matrix, only if the full matrix is FALSE (i.e. 0), then the result is
considered FALSE. Any non-zero element makes it true.
Note that FALSE = 0, TRUE = 1, and true is any non-zero value

Conditional statements: Case

Alternative way, if you know what values i can take on:

Listing 15: oxswitch.ox
switch_single (i)

{

case 0:

println ("zero"); // Single statement

case 1:

{

println ("one"); // Single compound statement

println ("So I said , one ...");

}

default:

println ("something else");

}

Conditional statements

Conditional statements: Assignment

Also possible:

A= Condition .? Value if true .: Value if false

Listing 16: oxcond.ox
dY= (dX > 0) ? 1 : -1; // One check only , scalar

mY= (mX .> 0) .? 1 .: -1; // Multiple elements at once

Very useful in creating dummies, think of probit models.

Missings

Further topics: NaN
Not a Number, or NaN for short is the missing value which is
supported by computer hardware.

I Use .NaN to represent the missing value in Ox code.
I In a matrix constant, you may use a dot to represent a NaN.
I Or use the predefined constant M NAN (defined in oxfloat.h).
I The format used when printing output is .NaN.

#include <oxfloat.h> // defines M_NAN

main()

{

decl mX, d1, d2;

mX = < . >; d1 = .NaN; d2 = M_NAN;

print(mX + 1, d1 == .NaN , " ", d2 / 2);

}

Any computation involving a NaN results in a NaN, so in this
example d2 / 2 is also .NaN. Comparison is allowed and d1 ==

.NaN evaluates to one (so is TRUE).
Preferably use ismissing(d1) or isdotmissing(mX) instead.

Further topics: NaN II

Functions operating with missings:

I deleter(mX): deletes all rows which have a NaN,

I selectr(mX): selects all rows which have a NaN,

I isdotnan(mX): returns matrix of 0’s and 1’s: 1 if the element
is a NaN, 0 otherwise,

I isnan(mX): returns 1 if any element is a NaN, 0 otherwise.

I isdotmissing(mX): returns matrix of 0’s and 1’s: 1 if the
element is a NaN or ± infinity, i.e. M_NAN, M_INF or
M_INF_NEG, 0 otherwise.

I ismissing(mX): returns 1 if any element is a NaN or ±
infinity, i.e. M_NAN, M_INF or M_INF_NEG, 0 otherwise.

Some constants

Using #include <oxfloat.h> delivers the constants

M_PI π
M_2PI 2π
M_PI_2 π/2
M_1_PI 1/π
M_SQRT2PI

√
2π

M_NAN Missing, test using isnan/ismissing
M_INF ∞, test using isdotinf/ismissing
M_INF_NEG −∞, test using isdotinf/ismissing

Exit

To exit Ox before reaching the end of the program, use

exit(iErr);

where iErr is an integer, the exit code Ox will return to the
operating system.

Syntax

Scope

Further topics: Scope

Any variable is available only within the block in which it is
declared.
In practice:

1. A local decl mX is only known below the declaration, within
the present set of braces { }

2. The argument to a function, e.g. mX in fnPrint(const mX),
is available within this one function

3. A global/static variable, with static decl s_vY, is available
in any code below, if this static declaration is made outside
any function

Syntax

Scope

Further topics: Scope II

Listing 17: oxglobal.ox
#include <oxstd.h>

#import <maximize >

static decl s_dNu; // Global , for use in probtL

probtL(const dX , const adP , const avScore , const amHess)

{

adP [0]= probt(dX, s_dNu );

return !ismissing(adP [0]);

}

main()

{

...

println ("Calculating the derivative of Student -t(x=", dX, ", df= ", dNu , ") density");

s_dNu= dNu; // Assign global at last possible moment

ir= Num1Derivative(probtL , dX, &dDX);

}

Rules for globals:
I Only use them when absolutely necessary (dangerous!)
I Declare them statically, at beginning of file
I Annotate them, s_
I Fill them at last possible moment
I Do not change them afterwards (unless absolutely necessary) 47/66

Syntax

Scope

Further topics: Scope III

Listing 18: oxscope.ox
#include <oxstd.h>

static decl s_vY; // Global/static variable available throughout this file

fnPrint(const mX)

{

decl vY; // Only available in fnPrint () block

vY= 4;

print ("In fnPrint. vY: ", vY , ", Static s_vY: ", s_vY , ", mX: ", mX);

}

main()

{

decl i, vY; // Only available in main () block

vY= 6;

for (i= 0; i < 5; ++i)

{

decl mX; // Only available within for loop

mX= rann(1, i+1);

println ("i= ", i, ", mX= ", mX);

}

s_vY= 2; // Fill global variable

fnPrint(vY);

}

Q: Ugly, confusing, incorrect use of Hungarian notation (where?)!

Syntax

Scope

Further topics: Scope IV

Some hints:

I Put a single decl at the top of each function (exception,
advanced: Parallel programming, local matrices)

I Pass arguments to functions if you need.

I Use no static/global variables, too dangerous

I Alternatively, instead of static variables: Use lambda
functions, to pass those extra variables. Much better!

49/66



PPEctr-Syntax

Syntax

Lambda

Further topics: Lambda functions
A novelty in Ox7: Anonymous or Lambda functions, useful for
getting rid of globals.

Listing 19: oxlambda.ox
#include <oxstd.h>

fnPrint(const mX, const vZ)

{

decl vY; // Only available in fnPrint () block

vY= 4;

print ("vY: ", vY, ", mX: ", mX, ", vZ: ", vZ);

}

main()

{

decl vY, vZ, fnPrintLambda;

vY= 6; vZ= 5;

fnPrintLambda= [=] (const mX) // Define lambda function , passing

{ // along vZ to fnPrint

fnPrint(mX, vZ);

};

fnPrintLambda(vY);

}

Can you predict the outcome of this program?

Syntax

Lambda

Lambda functions II

Lambda functions:

I allow you to hide arguments

I are very useful if some function is prespecified to have only a
certain number of fixed arguments (think: optimisation
through MaxBFGS, integration through quadpack)

I are defined in standard format
fnLambda= [=] (arguments to lambda function)

{

// effective code called

};

I cannot change local variables (in the example above: vZ

cannot be changed within the fnPrintLambda function, see
oxlambda err.ox)

Data selection

Data selection

Standard situation

I Large dataset of observations, many individuals
I Only those

I older than 18,
I female,
I with an income between 25-50k

should go into the sample...

Three words: selectifr, deleteifr, vecrindex

Some others: thinc, range

Data selection

Data: selectifr

First two either select or delete specific rows from a matrix.

Listing 20: seldata/seldata.ox
main()

{

...

// Load data including names

mX= loadmat("../ data/duration.in7", &asNames );

// Find out what are indices in data set

vSel= strfind(asNames , {"Age", "Sex", "Income"});

mX= selectifr(mX, mX[][ vSel [0]] .> 18); // Older than 18

mX= selectifr(mX, !mX[][ vSel [1]]); // Only female

mX= selectifr(mX, (mX[][ vSel [2]] .>= 25000) .&&

(mX[][ vSel [2]] .<= 50000)); // 25k -50k income

...

}

selectifr(mA, mB): Select only those rows of mX which have (at
least one) non-zero element in the corresponding rows of the
second argument.

Data selection

Data: deleteifr

You could also do it the other way around:

Listing 21: seldata/deldata.ox
main()

{

...

// Load data including names

mX= loadmat("../ data/duration.in7", &asNames );

// Find out what are indices in data set

vSel= strfind(asNames , {"Age", "Sex", "Income"});

mX= deleteifr(mX, mX[][ vSel [0]] .<= 18); // Not older than 18

mX= deleteifr(mX, mX[][ vSel [1]]); // Not female

mX= deleteifr(mX, (mX[][ vSel [2]] .< 25000) .||

(mX[][ vSel [2]] .> 50000)); // <25k or >50k income

...

}

Data selection

Data: vecrindex

For more extensive situations: Use vecrindex?

Listing 22: seldata/inddata.ox
// Create vector of zero/ones , indicating observations to use

vI= (mX[][ vSel [0]] .> 18) .&&

!mX[][ vSel [1]] .&&

(mX[][ vSel [2]] .>= 25000) .&&

(mX[][ vSel [2]] .<= 50000);

// Re -using vSel , to indicate observations to use}

vSel= vecrindex(vI);

print ("Limits of indexed observations: ", limits(mX[vSel ][]));

// Or , old tricks:

mX= selectifr(mX, vI);

I These functions are used extensively in Initialise()

I Warning: Don’t do this within likelihood function...

I Changing big data matrices, sorting, whatever: Relatively slow

Data selection

Data: thinr/thinc

If a matrix is too large to handle, thinr can select a number of rows
from a large matrix:

mXSmall= thinr(mX, 1000); // Get 1000 rows from mX , if possible

A similar effect, with more control, can be obtained using range:

iN= rows(mX);

vI= range(0, iN -1, 100);

mXSmall= mX[vI][]; // Get row 0, 100, 200, ... from mX

Graphics

A package: OxDraw (or GnuDraw...)

Ox graphics are displayed within OxMetrics. Needs the professional
version for Windows.

Alternatively, use GnuDraw: Displays graphics in GnuPlot on
Windows, OSX, Unix. Compatible in usage, easy to switch.

Listing 23: stack/drawstack.ox
#include <oxdraw.h>

// #include <packages/gnudraw/gnudraw.h> // Alternatively

// Draw stackloss regressors on Y, stackloss itself on X axis

DrawXMatrix (0, mX’, asXVar , vY’, sYVar , 1, 2);

SaveDrawWindow("graphs/stackloss.eps");

ShowDrawWindow ();

From the manual:
DrawXMatrix(const iArea , const mYt , const asY , const vX, const sX, ...);

DrawXMatrix(const iArea , const mYt , const asY , const vX,

const sX , const iSymbol , const iIndex );

Graphics

OxDraw (or GnuDraw...) II
I Graphing appears in graphing area, first argument
I Draws rows at a time
I Puts in a label. For multiple Y-values, give an array of labels

{"yHat", "y", "cons"}

I Can draw XY data, time series data, densities, QQ-plots etc.
I Takes extra arguments specifying line types, colours etc.
I After drawing the graph, and before showing it, the last

graphing command can be adjusted using DrawAdjust(...)

I For daily time series data, use e.g.
DrawTMatrix(iArea, mY, asYVar, vDates, 0, 0);

I Save the graphics in eps, pdf or gwg format (oxdraw), or also
plb, png, tex and others (gnudraw)

I Can show the graph on the screen (professional version of Ox)
I Close the graph if necessary before continuing

Parallelisation

Possibilities

Parallel programming and speed

Using multiple CPUs can speed up your calculations, but involve
complications. Standard situations

I Run one program per CPU, wait for results (slowest option,
obviously, but simple and easy to debug)

I Run one program on x CPUs, and Ox will automatically do
matrix operations in parallel (very simple, could deliver decent
speed-up for matrix computations)

I Run one program on x CPUs, explicitly indicating where Ox
may parallelize (involves some change to your program, may
lead to hard-to-track errors/race conditions, good speed-up)

I Run one program per CPU, multiple programs simultaneously
(good for simulations, possibly hard to combine results, track
random numbers generated, optimal speed-up)

Parallelisation

Multiple CPUs, matrix computations

Speedup of using multiple CPUs

Run simple matrix calculations on either 1 or 8 CPUs:

Listing 24: speed matrix.ox
for (r= 0; r < iR; ++r)

mXtX= mX’mX;

with output

oxl -rp1 speed_matrix.ox

Ox Professional version 7.09 (Linux_64/MT) (C) J.A. Doornik , 1994 -2014

Time spent in routines

Matrix 4.16 1.00

Total: 4.15

oxl -rp8 speed_matrix.ox

Ox Professional version 7.09 (Linux_64/MT) (C) J.A. Doornik , 1994 -2014

Time spent in routines

Matrix 1.61 1.00

Total: 1.61

Speedup: Factor 2.6, for using 8 CPUs instead of 1.

Parallelisation

Parallel loops

Parallel loops

Alternatively, one can use a parallel loop, e.g.

parallel for (i= 0; i < iK; ++i)

for (j= 0; j <= i; ++j)

{

for (k= 0; k < iN; ++k)

mXtX[i][j]+= mX[k][i] * mX[k][j];

mXtX[j][i]= mXtX[i][j];

}

Hence, only change to code: add word parallel...

Parallelisation

Parallel loops

Parallel loops II

for (i= 0; i < iK; ++i) // Outer loop

for (j= 0; j <= i; ++j) // Middle loop

{

for (k= 0; k < iN; ++k) // Inner loop

mXtX[i][j]+= mX[k][i] * mX[k][j];

mXtX[j][i]= mXtX[i][j];

}

Possibilities for parallelisation here:

I No parallel loop: Good choice if parallelisation is best done at
the matrix level

I Outer loop: Otherwise best choice, little overhead in
parallelisation

I Middle loop: More overhead, slower

I Inner loop: Incorrect, no warning! (Q: Why?)

Parallelisation

Parallel loops

Parallel loops III

Possibilities for parallelisation here:

I ...

I Inner loop: Incorrect, no warning!

I Or use matrix multiplications, far quicker...

Listing 25: oxl -rp8 speed loop parallel.ox
Ox Professional version 7.09 (Linux_64/MT) (C) J.A. Doornik , 1994 -2014

Time spent in routines

Non -parallel loop 2.44 0.53

Parallel outer loop 0.75 0.16

Parallel middle loop 1.40 0.30

Matrix 0.01 0.00

Total: 4.59

Parallelisation

Parallel loops

Parallel Loops: Race conditions

What if one parallelises the inner loop?

Listing 26: speed parallel loop err.ox
for (i= 0; i < iK; ++i) // Outer loop

for (j= 0; j <= i; ++j) // Middle loop

{

parallel for (k= 0; k < iN; ++k) // Inner loop

mXtX[i][j]+= mX[k][i] * mX[k][j];

mXtX[j][i]= mXtX[i][j];

}

Multiple threads try to update mXtX[i][j] at the same time ⇒
Incorrect final answer...

Own responsibility ... no checks through Ox ...
parallelisation can be dangerous ...

Parallelisation

Parallel loops

Parallel loops: Rules of thumb
When running code in parallel,

I n threads are created;
I each thread gets its own set of variables
I all integers/doubles are copied (local copies)
I all other variables are shared between threads
I and after the parallel section, all local copies of

integers/doubles disappear again

Hence
I be careful to only update matrix elements in separate threads
I do not change global variables/matrices
I it may be useful to call a function within the parallel loop,

such that you make very clear what information is available
within the loop, what may be changed.

Anyhow, check manual thoroughly, and take care...

Parallelisation

Parallel loops

More syntax?

I Check manual in html, on your own machine or online at
http://doornik.com/ox/

I Check Ox professional manual Ox.pdf (only with professional
version of Ox)

I Check with colleagues

I ... and no, checking with Google does not tend to help for
this language...

